Competitive Labor Market

```
> restart;
```

```
> Production := {q = g[1]*L-g[2]*L*L};
```

$$Production := \left\{q = g_1\,L - g_2\,L^2\right\} \quad \textbf{(1.1)}$$

```
> Marginal_Product := {MPL = diff(subs(Production,q),L)};
```

$$Marginal\_Product := \{MPL = g_1 - 2\,g_2\,L\} \quad \textbf{(1.2)}$$

```
> Revenue := {R=p*q};
```

$$Revenue := \{R = p\,q\} \quad \textbf{(1.3)}$$

```
> MarginalRevenueProduct := {MRPL=diff(subs(Revenue,Production,
  R),L)};
```

$$MarginalRevenueProduct := \{MRPL = p\,(g_1 - 2\,g_2\,L)\} \quad \textbf{(1.4)}$$

```
>  {w=MRPL};
```

$$\{w = MRPL\} \quad \textbf{(1.5)}$$

```
> LaborDemand[firm] := subs(MarginalRevenueProduct,%);
```

$$LaborDemand_{firm} := \{w = p\,(g_1 - 2\,g_2\,L)\} \quad \textbf{(1.6)}$$

```
> employment[firm] := subs(L=L[f],solve(%,{L}));
```

$$employment_{firm} := \left\{L_f = \frac{1}{2}\,\frac{-w + p\,g_1}{p\,g_2}\right\} \quad \textbf{(1.7)}$$

```
> Employment[industry] := { L = N*L[f]};
```

$$Employment_{industry} := \{L = N\,L_f\} \quad \textbf{(1.8)}$$

```
> subs(employment[firm],Employment[industry]);
```

$$\left\{L = \frac{1}{2}\,\frac{N\,(-w + p\,g_1)}{p\,g_2}\right\} \quad \textbf{(1.9)}$$

```
> LaborDemand[industry] := solve(%,{w});
```

$$LaborDemand_{industry} := \left\{w = -\frac{(2\,g_2\,L - N\,g_1)\,p}{N}\right\} \quad \textbf{(1.10)}$$

```
> data := {N=100,p=10,a[m]=80,b[m]=1/40,a[f]=-20,b[f]=1/10, g
  [1]=30,g[2]=1/10};
```

$$data := \left\{N = 100,\ p = 10,\ a_m = 80,\ b_m = \frac{1}{40},\ a_f = -20,\ b_f = \frac{1}{10},\ g_1 = 30,\ g_2 = \frac{1}{10}\right\} \quad \textbf{(1.11)}$$

```
> subs(data,LaborDemand[industry]);
```

$$\left\{w = -\frac{1}{50}\,L + 300\right\} \quad \textbf{(1.12)}$$

```
> Supply[men] := {w[m]=a[m]+b[m]*L[m]};
```

$$Supply_{men} := \{w_m = a_m + b_m\,L_m\} \quad \textbf{(1.13)}$$

```
> Supply[women] := {w[f]=a[f]+b[f]*L[f]};
```

(1.14)

$$Supply_{women} := \{w_f = a_f + b_f L_f\} \quad (1.14)$$

```
> Supply[both] := {L=L[f]+L[m]};
```

$$Supply_{both} := \{L = L_f + L_m\} \quad (1.15)$$

```
> Employment[men] := solve(subs(w[m]=w,Supply[men]),{L[m]});
```

$$Employment_{men} := \left\{L_m = \frac{w - a_m}{b_m}\right\} \quad (1.16)$$

```
> Employment[women] := solve(subs(w[f]=w,Supply[women]),{L[f]})
  ;
```

$$Employment_{women} := \left\{L_f = \frac{w - a_f}{b_f}\right\} \quad (1.17)$$

```
> LaborSupply[industry] := solve(subs(Employment[men],
  Employment[women],Supply[both]),{w});
```

$$LaborSupply_{industry} := \left\{w = \frac{L\, b_f b_m + a_f b_m + a_m b_f}{b_m + b_f}\right\} \quad (1.18)$$

```
> subs(data,LaborSupply[industry]);
```

$$\left\{w = \frac{1}{50}\, L + 60\right\} \quad (1.19)$$

```
> Equilibrium := solve(`union`(LaborDemand[industry],
  LaborSupply[industry]),{w,L});
```

$$Equilibrium := \left\{L = \frac{N\,(b_m\, p\, g_1 - a_f b_m - a_m b_f + b_f\, p\, g_1)}{N\, b_f b_m + 2\, b_m\, p\, g_2 + 2\, b_f\, p\, g_2},\, w = \frac{p\,(2\, g_2\, a_f b_m + 2\, g_2\, a_m b_f + g_1 N\, b_f b_m)}{N\, b_f b_m + 2\, b_m\, p\, g_2 + 2\, b_f\, p\, g_2}\right\} \quad (1.20)$$

```
> subs(data, Equilibrium);
```

$$\{L = 6000,\, w = 180\} \quad (1.21)$$

```
> plots[implicitplot](subs(data,LaborSupply[industry]),L=3000.
  .10000,w=100..300,title="Supply",thickness=3, font=
  [Helvetica,bold,14],titlefont=[Helvetica,bold,16]);
> ps := %:
```




```
> plots[implicitplot](subs(data,LaborDemand[industry]),L=3000.
  .10000,w=100..300,title="Demand",thickness=3,font=[Helvetica,
  bold,14],titlefont=[Helvetica,bold,16]);
> pd := %:
```




```
> plots[display]([ps,pd],title="Equilibrium");
```



>

>

>

Labor Supply Details

>

```
> Supply[both] := {L=max(L[f],0)+max(L[m],0)};
```

$$Supply_{both} := \{L = \max(0, L_f) + \max(0, L_m)\}$$ **(2.1)**

```
> subs(Employment[men],Employment[women],data,Supply[both]);
> plots[implicitplot](%,L=000..8000,w=0..300,title="Supply",
  thickness=3, font=[Helvetica,bold,14],titlefont=[Helvetica,
  bold,16],numpoints=50000);
> ps2 := %:
```

$$\{L = \max(0, 10\,w + 200) + \max(0, 40\,w - 3200)\}$$




```
> plots[display]([ps2,pd],title="Equilibrium");
```



Monopsony

```
> LaborCost := {LC[m] = w[m]*L[m],LC[f]=w[f]*L[f]};
```

$$LaborCost := \{LC_m = w_m L_m, LC_f = w_f L_f\} \quad (3.1)$$

```
> subs(Supply[men],Supply[women],LaborCost);
```

$$\{LC_m = (a_m + b_m L_m) L_m, LC_f = (a_f + b_f L_f) L_f\} \quad (3.2)$$

```
> MarginalLaborCost := {MLC[m] = diff(subs(%,LC[m]),L[m]),MLC
  [f] = diff(subs(%,LC[f]),L[f])};
```

$$MarginalLaborCost := \{MLC_m = 2\, b_m L_m + a_m, MLC_f = 2\, b_f L_f + a_f\} \quad (3.3)$$

```
> MarginalRevenueProduct;
```

$$\{MRPL = p\, (g_1 - 2\, g_2 L)\} \quad (3.4)$$

```
> ProfitMax := {MLC[m]=MRPL,MLC[f]=MRPL};
```

$$ProfitMax := \{MLC_m = MRPL, MLC_f = MRPL\} \quad (3.5)$$

```
> subs(L=(L[m]+L[f])/N,MarginalRevenueProduct);
```

(3.6)

$$\left\{MRPL = p\left(g_1 - \frac{2\,g_2\,(L_m + L_f)}{N}\right)\right\} \quad (3.6)$$

```
> eval(subs(%,MarginalLaborCost,ProfitMax));
```

$$\left\{2\,b_m\,L_m + a_m = p\left(g_1 - \frac{2\,g_2\,(L_m + L_f)}{N}\right), 2\,b_f\,L_f + a_f = p\left(g_1 - \frac{2\,g_2\,(L_m + L_f)}{N}\right)\right\} \quad (3.7)$$

```
> MonopsonyEmployment := solve(%,{L[f],L[m]});
```

$$MonopsonyEmployment := \left\{L_f = \frac{1}{2}\,\frac{-N\,b_m\,a_f + N\,b_m\,p\,g_1 - p\,g_2\,a_f + p\,g_2\,a_m}{b_f\,p\,g_2 + b_m\,p\,g_2 + N\,b_f\,b_m}, L_m = \frac{1}{2}\,\frac{-b_f\,N\,a_m + p\,g_2\,a_f - p\,g_2\,a_m + p\,g_1\,N\,b_f}{b_f\,p\,g_2 + b_m\,p\,g_2 + N\,b_f\,b_m}\right\} \quad (3.8)$$

```
> Sol[Monopsony] := eval(subs(data,%));
```

$$Sol_{Monopsony} := \{L_f = 1200, L_m = 2800\} \quad (3.9)$$

```
> subs(Sol[Monopsony],data,{Supply[men],Supply[women]});
```

$$\{\{w_m = 150\}, \{w_f = 100\}\} \quad (3.10)$$

```
> eval(subs(data,Sol[Monopsony],MarginalLaborCost));
```

$$\{MLC_m = 220, MLC_f = 220\} \quad (3.11)$$

```
> eval(subs(L=(L[m]+L[f])/N,data,Sol[Monopsony]
,MarginalRevenueProduct));
```

$$\{MRPL = 220\} \quad (3.12)$$

```
> subs(L=(L[m]+L[f])/N,data,Sol[Monopsony],Production);
```

$$\{q = 1040\} \quad (3.13)$$

```
> subs(%,data,Q=N*q);
```

$$Q = 104000 \quad (3.14)$$

```
>
```